///////////////////////////////////////////////////
// Election Administration Polarization
// Main RD Table
///////////////////////////////////////////////////

* Set the working directory
gl path "~/Library/CloudStorage/GoogleDrive-danmckinleythompson@gmail.com/My Drive/ElecAdminPolarization/How_Partisan_Is_Local_Election_Admin_Replication"

* Bring in the presidential election analysis data
use "$path/analysis_data/rd_election_analysis_data.dta", clear
keep if office=="pres"

* Use CCT's automated bandwidth selection procedure
rdbwselect r_oos_state_year_vs_dem rv, vce(cluster election_id) kernel(uniform)
gl bw = e(h_mserd)

* Spec 1, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") new("B") tau(0.02) alpha(0.05) onesided

* Spec 2, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*2, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 3, Residualized Outcome
reg r_oos_state_year_vs_dem treat rv treat_rv if abs(rv)<$bw*0.5, vce(clust election_id)
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("reg") old("B") tau(0.02) alpha(0.05) onesided

* Spec 4, Residualized Outcome
rdrobust r_oos_state_year_vs_dem rv, vce(cluster election_id) all
rdtab r_oos_state_year_vs_dem treat rv election_id, est_command("rdrobust") old("B") tau(0.02) alpha(0.05) onesided

qui {
cap log close
set linesize 255
log using "$path/output/rd_table_elec.tex", text replace
noi di "\begin{table}[t]"
noi di "\centering"
noi di "\caption{\textbf{Effect of Democratic Election Officials on Democratic Presidential Vote Share.}"
noi di "\label{tab:rd_table_elec}}"
noi di "\begin{tabular}{lcccc}"
noi di "\toprule \toprule"
noi di " & \multicolumn{4}{c}{Dem Pres Vote Share}\\"
noi di " & (1) & (2) & (3) & (4) \\"
noi di "\midrule"
noi di "Dem Elec Official & " %4.3f B[1,1] " & " %4.3f round(B[1,2], 0.001) " & " %4.3f B[1,3] " & " %4.3f B[1,4] "\\"
noi di " & (" %4.3f B[2,1] ") & (" %4.3f B[2,2] ") & (" %4.3f B[2,3] ") & (" %4.3f B[2,4] ") \\[2mm]"
noi di "\midrule"
noi di "N & " %4.0f B[3,1] " & " %4.0f B[3,2] " & " %4.0f B[3,3] " & " %4.0f B[3,4] " \\ "
noi di "Bandwidth & " %4.2f B[5,1] " & " %4.2f B[5,2] " & " %4.2f B[5,3] " & " %4.2f B[5,4] " \\ "
noi di "Bandwidth Selection & CCT & CCT*2 & CCT/2 & CCT \\"
noi di "Kernel & Uniform & Uniform & Uniform & Triangular \\"
noi di "Min Detectable Effect  & " %4.3f B[7,1] " & " %4.3f B[7,2] " & " %4.3f B[7,3] " & " %4.3f B[7,4] " \\ "
noi di "\bottomrule \bottomrule"
noi di "\multicolumn{5}{p{.72\textwidth}}{\footnotesize Robust standard errors in parentheses. "
noi di "The outcome is first regressed on a state- and year-specific lag using all counties including those for which clerk election "
noi di "results are not available. The regression discontinuity is estimated using the residuals from that regression. "
noi di "The bandwidth row reports the number of maximum clerk win margin allowed for inclusion in each specificaiton. "
noi di "CCT refers to \cite{calonico2014robust} bandwidth selection procedure."
noi di "Min detectable effect refers to the minimum effect that a one-sided test with a 0.05 alpha would have 80\% power to detect.}"
noi di "\end{tabular}"
noi di "\end{table}"
log off
}
